

Super-Install: Das Aus für

Nahezu jeder C 64-Besitzer kennt das Problem: Die euphorische Begeisterung über den neuen Drucker mit passendem Interface legt sich oft beim ersten Probeausdruck. Der C 64 und der angeschlossene Drucker verstehen sich nicht.

Manche Textverarbeitungsprogramme besitzen zum Glück eine eigene Druckeranpassung. Doch was tun, wenn dies nicht der Fall ist? Entweder man versucht es »zu Fuß«, indem man selbst eine entsprechende Routine schreibt, oder aber — man kann auf Super-Install zurückgreifen. Dieses komplett in Basic geschriebene Programm wird im folgenden am Beispiel einer Textverarbeitung vorgestellt und beschrieben.

Spaziergang durch den Wertedschungel

Nur in seltenen Fällen weiß der Anwender, welche Werte das Textprogramm dem Drucker schickt, um das gerade anstehende Zeichen auszugeben. Die meisten Programme haben mit dem amerikanischen Standardzeichensatz keinerlei Probleme, doch wenn es darum geht, Umlaute auszugeben, spielen die Drucker des öfteren verrückt. Um eine solide Grundlage für das Arbeiten mit Super-Install zu schaffen, sollte man sich als erstes eine Liste der ASCII-Werte seines Druckers erstellen. Dies geschieht mit folgendem kleinen Programm:

```
10 OPEN 1,4,7
20 FOR I=0 TO 255 : PRINT I
30 POKE 198,0 : WAIT 198,1
   : GET X$: IF
   X$=CHR$(13) THEN NEXT
40 PRINT #1,CHR$(I); I
50 NEXT
60 CLOSE
```

Wenn Sie <RETURN> drücken, wird das Zeichen übersprungen, ansonsten erfolgt die Ausgabe auf dem

Drucker mit Wert und Bedeutung (Bild 1). Nachdem diese Wertetabelle erstellt worden ist, laden Sie als nächstes Ihr Textverarbeitungsprogramm. Dort erfassen Sie nun alle sinnvollen Zeichen (Bild 2) und drucken diese aus. Sollten Sie nach einem Vergleich von Eingabe und Ausdruck (Bild 3) keine Unterschiede feststellen, brauchen Sie keine Anpassung durchführen. Ansonsten tritt jetzt Super-Install in Aktion. Alle nicht übereinstimmenden Zeichen werden der Übersicht halber in eine Liste (Bild 4) eingetragen. Die verschiedenen Werte der Zeichen werden ebenfalls in die Liste geschrieben. Man stellt fest, daß viele Zeichen mehrfach belegt sind. Nun könnte das große Rätselraten beginnen, wenn man nicht etwas logisch vorgehen würde. Es empfiehlt sich, zuerst die Zeichen zu ersetzen, die in der Wertetabelle nur einmal vorkommen. In diesem Fall sind das die Zeichen »Ä« mit dem Wert 219, »Ö« mit 220, »Ü« mit 221 und »A« mit 193. Man sollte grundsätzlich nicht versuchen, die gesamte Anpassung auf einmal durchzuführen, da man schnell den Überblick verlieren kann.

Zunächst werden in »Super-Install« die Werte
ä = 91 durch 219
ö = 92 durch 220
ü = 93 durch 221
a = 65 durch 193
ersetzt. Führt dies nicht zum Erfolg, so hat man, wie aus der Wertetabelle ersichtlich, noch eine andere Möglichkeit:
ä = 219 wird durch 251
ö = 220 durch 252
ü = 221 durch 253
a = 193 durch 225
ersetzt.

Wenden wir uns nun den anderen Zeichen zu. Bei den verbleibenden Zeichen ist folgendes zu beachten: die Zeichen »;« (=) !« müssen ja selbst auch gedruckt werden. Wie wir gesehen haben, werden diese Zeichen richtig zu Papier gebracht. Der Wert darf also nicht

Vergessen Sie die ewigen Probleme mit der Druckeranpassung. Super-Install versetzt Sie in die Lage, beliebige ASCII-Werte zu ändern und sorgt dafür, daß die entsprechenden Zeichen auch wie gewünscht auf dem Drucker erscheinen.

überschrieben werden. Man erhält zum Beispiel an Stelle des »;« ein »ä«.

Drucker überlistet

In den meisten Fällen kann man davon ausgehen, daß der niedrigste Wert für das Zeichen selbst reserviert ist und demzufolge nicht umdefiniert werden sollte.

Sehen wir uns zunächst die Werte für »ä ö ü ß« näher an. Die ersten Werte, so bleibt zu vermuten, scheiden von vornherein aus. Wenn man unterstellt, daß die Werte vom Textprogramm nicht kreuz und quer verteilt sind, sondern nacheinander stehen, so können einige Werte schon mal unter den Tisch fallen.

Mit Super-Install werden zunächst die Werte
; = 123 durch 91
(= 124 durch 92
= = 125 durch 93
) = 126 durch 94
ersetzt. Führt dies wiederum nicht zum Erfolg, ist es sinnvoll, die Anpassung mit folgenden Werten zu wiederholen:

; = 91 wird durch 187
(= 92 durch 188
= = 93 durch 189
) = 94 durch 190
ersetzt. Weitere Kombinationsmöglichkeiten sind möglich, aber nicht erforderlich. Wenden wir uns dem letzten Zeichen zu. Aus den eben genannten Gründen setzt man nicht den Wert 33 für das »!«, sondern den Wert 97. Nun ist es unlogisch, wie bisher den ersten Wert (65) einzusetzen, also für »!« = 97 die 65. Es wurde bereits für den Wert 65 die 193 eingesetzt, was zur Folge haben würde, daß nun beim Drucken des »a« ein »A« entstünde (den Vorgang muß man erst

einmal langsam durchdenken). Geht man nicht logisch vor, so erkennt man spätestens beim Ausdruck, daß die nächste Kombination zum Erfolg führt.

Man ersetzt »!« = 97 durch 225.

Soviel vorerst zur Anpassung, wobei noch anzumerken ist, daß Super-Install die Umlaute und Sonderzeichen natürlich nicht anzeigt, sondern nur die dem Commodore-Code entsprechenden Zeichen.

Hoher Bedienungskomfort

Im folgenden werden nun der Programmaufbau und die Bedienung von Super-Install näher erläutert. Wie bereits erwähnt, ist Super-Install ein Basic-Programm. Es wird also ganz gewöhnlich mit

LOAD "Super-Install",8
von Diskette geladen. Nach dem Starten erscheint eine übersichtliche Bildschirmmaske, in der die ersten Werte mit Zeichenbedeutung zu finden sind. Mit den Cursor-tasten kann man in der Tabelle blättern. Um nun einen Wert zu ändern, bewegt man den Cursor neben das zu ändernde Zeichen und gibt den neuen Wert ein. Sofort erscheint die neue Belegung. Nachdem alle erforderlichen Definitionen durchgeführt sind, wird die neue Wertetabelle abgespeichert. Dazu muß der Anwender mit der Taste <S> den Menüpunkt Sichern auswählen. Das Programm meldet sich mit »CODE SICHERN:«. An dieser Stelle geben Sie den frei wählbaren Namen Ihrer Wertetabelle ein, der sich sinnvollerweise auf das angewendete

widerspenstige Drucker

Programm beziehen sollte. Danach verlangt Super-Install die Eingabe der Adresse. Und damit wären wir dann bei der eigentlichen Besonderheit von Super-Install. Hier wird mit Hilfe des Basic-Programms ein Maschinensprache-Programm erstellt, wofür Super-Install allerdings die Startadresse benötigt. Nachdem die Adresse erfaßt ist, verlangt Super-Install noch den Namen des Textprogramms, das unterstützt werden soll. Natürlich können Sie hier auch eigene Programme angeben, da die Maschinenroutine auch mit Basic-Programmen zusammenarbeitet.

Wenn alle Angaben ordnungsgemäß gemacht wurden, kann nun mit <CTRL+X> der Lader für das Programm erstellt werden. Dieser erhält den Namen »LADER.Textprogramm«.

Wenn das Anwenderprogramm mit Druckeranpassung gestartet werden soll, lädt man zuerst den Lader in den Speicher. Alles andere übernimmt dann der C 64.

Längst ist jetzt eine nähere Beschreibung der erstellten Maschinenroutine fällig. Die Routine muß unbedingt in einen Speicherbereich geladen werden, der vom eigentlichen Anwenderprogramm nicht belegt wird. Allerdings kann der Kassettenpuffer dafür nicht verwendet werden, da die Routine zu lang ist. Am besten verwendet man den Speicherbereich oberhalb von Adresse 49152. Noch etwas muß unbedingt beachtet werden: Die Geräteadresse des Druckers muß 4 sein, da in der Maschinenroutine sonst, ohne das Zeichen zu konvertieren, in die BSOUT-Routine des C 64 gesprungen wird. Doch nun zu der Maschinenroutine (Bild 5) im einzelnen. Als erstes wird dort der OUTPUT-Vektor verbogen, so daß er auf die eigentliche Änderungsroutine zeigt. Steht ein Zeichen zur Ausgabe an, wird als erstes abgeprüft, ob die Geräteadresse auf 4 steht. Ist

dies nicht der Fall, wird das Zeichen auf den Bildschirm ausgegeben und nicht konvertiert. Andernfalls wird aus der direkt hinter dem Maschinenprogramm gespeicherten Wertetabelle, mit dem übergebenen Wert im Y-Register indiziert, der

entsprechende Wert geholt und im Akkumulator an die Ausgaberroutine BSOUT übergeben. Das war dann auch schon alles. Im Prinzip kein großer Aufwand, aber es erfüllt in jedem Fall seinen Zweck. Der, wie bereits erwähnt, größte Vorteil, liegt in

der Anwendbarkeit auf fast alle Programme. Dies soll an einem kleinen Beispiel demonstriert werden. Dazu nehmen Sie wieder das kleine Programm zum Ausdrucken der ASCII-Werte. Die alte Liste können Sie ohne weiteres verwenden. Mit Super-Install wird nun der Wert für »a« durch den Wert von »Z« ersetzt und umgekehrt. Das Ganze speichern Sie unter dem Namen »Test« ab. Als Adresse gibt man 49152 an.

Für den Namen des Textverarbeitungsprogramms wird einfach der Name des Programms zum Drucken der Wertetabelle erfaßt. Mit <CTRL+X> erstellen Sie sodann noch den Lader für den kleinen Test. Wenn mit dem Lader ein Basic-Programm aufgerufen werden soll, muß in Zeile 5 des Laders die 1 nach der Geräteadresse entfernt und das Programm nochmals abgespeichert werden. Nachdem dann »NEW« eingegeben wurde, kann es losgehen. Mit »LADER.Programmname« wird das Ladeprogramm eingeladen. Dieses wird dann nur noch mit »RUN« gestartet, und nach kurzer Zeit beginnt der Drucker mit seinem Werk. Beim Vergleichen der neuen Wertetabelle mit der alten, kann man feststellen, daß sich hier einiges geändert hat. So wurde, um auf obiges Beispiel zurückzukommen, das »a« mit dem »Z« vertauscht.

Zu guter Letzt...

Mit Super-Install hat auch der mit der Maschinenprogrammierung weniger vertraute Anwender die Möglichkeit, mit den Problemen der Druckeranpassung fertig zu werden. Noch dazu kann Super-Install sowohl bei der Textverarbeitung als auch bei eigenen Programmen verwendet werden, die Garantie für ein breites Anwendungsspektrum.

(Mario Schacht/rf)

```
1234567890+-
!"#$%&'()*

qwertyuiop*
QWERTYUIOP

asdfghjkl;=
ASDFGHJKL

zxcvbnm,./
ZXCVBNM?

äöüß ÄÖÜ
```

Bild 2. Text-Eingabe

```
1234567890+-
!"#$%&'()*

qwertyuiop*
QWERTYUIOP

!sdfghjkl;=
aSDFGHJKL

zxcvbnm,./
ZXCVMBNM?

; (=) äöü
```

Bild 3. Druck-Ausgabe

= 32	f = 70	, = 108	4 = 180	Z = 218
! = 33	g = 71	- = 109	5 = 181	Ä = 219
" = 34	h = 72	. = 110	6 = 182	Ö = 220
# = 35	i = 73	/ = 111	7 = 183	Ü = 221
\$ = 36	j = 74	0 = 112	8 = 184	ß = 222
% = 37	k = 75	1 = 113	9 = 185	_ = 223
& = 38	l = 76	2 = 114	: = 186	' = 224
' = 39	m = 77	3 = 115	; = 187	a = 225
(= 40	n = 78	4 = 116	(= 188	b = 226
) = 41	o = 79	5 = 117	= = 189	c = 227
* = 42	p = 80	6 = 118	? = 190	d = 228
+ = 43	q = 81	7 = 119	) = 191	e = 229
, = 44	r = 82	8 = 120	\$ = 192	f = 230
- = 45	s = 83	9 = 121	A = 193	g = 231
. = 46	t = 84	: = 122	B = 194	h = 232
/ = 47	u = 85	; = 123	C = 195	i = 233
0 = 48	v = 86	(= 124	D = 196	j = 234
1 = 49	w = 87	= = 125	E = 197	k = 235
2 = 50	x = 88	) = 126	F = 198	l = 236
3 = 51	y = 89	! = 161	G = 199	m = 237
4 = 52	z = 90	" = 162	H = 200	n = 238
5 = 53	ä = 91	# = 163	I = 201	o = 239
6 = 54	ö = 92	\$ = 164	J = 202	p = 240
7 = 55	ü = 93	% = 165	K = 203	q = 241
8 = 56	ß = 94	& = 166	L = 204	r = 242
9 = 57	_ = 95	' = 167	M = 205	s = 243
: = 58	= = 96	(= 168	N = 206	t = 244
; = 59	! = 97	) = 169	O = 207	u = 245
(= 60	" = 98	* = 170	P = 208	v = 246
= = 61	# = 99	+ = 171	Q = 209	w = 247
) = 62	\$ = 100	, = 172	R = 210	x = 248
? = 63	% = 101	- = 173	S = 211	y = 249
' = 64	& = 102	. = 174	T = 212	z = 250
a = 65	' = 103	/ = 175	U = 213	ä = 251
b = 66	(= 104	0 = 176	V = 214	ö = 252
c = 67	) = 105	1 = 177	W = 215	ü = 253
d = 68	* = 106	2 = 178	X = 216	ß = 254
e = 69	+ = 107	3 = 179	Y = 217	

Bild 1. Die verwendeten Druckerwerte

auf Bildschirm		auf Drucker	
Codes	Zel	Codes	Ze2
91, 251	ä	59, 123, 187	;
92, 252	ö	40, 60, 104, 124, 168, 188	(
93, 253	ü	61, 125, 189	=
94, 222, 254	ß	41, 62, 105, 126, 169, 190	)
219	Ä	91, 251	ä
220	Ö	92, 252	ö
221	Ü	93, 253	ü
65, 225	a	33, 97, 161	!
193	A	65, 225	a

Bild 4. Die Tabelle mit unterschiedlichen Zeichen

C000 = A9 0B	LDA #\$0B	; Output-Vektor auf \$C00B
C002 = 8D 26 03	STA \$0326	
C005 = A9 C0	LDA #\$C0	
C007 = 8D 27 03	STA \$0327	
C00A = 60	RTS	
C00B = 48	PHA	
C00C = A5 9A	LDA \$9A	
C00E = C9 04	CMP #\$04	; Ausgabe auf Drucker?
C010 = D0 0C	BNE \$C01E	
C012 = 68	PLA	
C013 = 8C 23 C0	STY \$C023	; Y-Register sichern
C016 = A8	TAY	
C017 = B9 25 C0	LDA \$C025,Y	; Zeichen aus Tabelle holen
C01A = 48	PHA	
C01B = AC 23 C0	LDY \$C023	; Y-Register holen
C01E = 68	PLA	
C01F = 4C CA F1	JMP \$F1CA	; BSOUT: Zeichen ausgeben

Bild 5. Der erzeugte Maschinencode

```

10 REM ***** <060>
20 REM ** -- SUPER-INSTALL -- ** <048>
30 REM ** M. SCHACHT / M.KLEMM ** <216>
40 REM ** BERLIN 51 (C) 1985 ** <106>
50 REM ***** <100>
60 POKE 53280,0:POKE 53281,0:POKE 53272,23
:PRINT CHR$(147)CHR$(159)CHR$(9) <235>
61 POKE 788,52:REM STOP BLOCKIEREN <117>
62 DIM LA$(11),DT(36),IN(255),BI(255),IN$(16) <179>
65 FOR I=0 TO 38:BL$=BL$+CHR$(32):NEXT <036>
67 PO=23:PT=10:PR$="... BITTE WARTEN ...": <046>
GOSUB 100
70 REM -----TABELLENWERTEVORGABE <126>
73 FOR I=0 TO 31:BI(I)=I+128:BI(I+32)=I+32
:BI(I+64)=I:BI(I+96)=I+64 <084>
74 BI(I+128)=I+192:BI(I+160)=I+96:BI(I+192)
)=I+64:BI(I+223)=I+95:NEXT:BI(255)=94 <009>
75 FOR I=0 TO 255:IN(I)=I:NEXT:POKE 631,17
:POKE 198,1 <128>
76 RESTORE:FOR I=0 TO 36:READ DT(I):NEXT <199>
77 FOR I=0 TO 11:READ LA$(I):NEXT <161>
78 ZE=10:N$="CODETABELLEN NAME":AD=49152:T
X$="TEXTPROGRAMM" <199>
80 GOTO 919 <121>
98 : <074>
99 REM -----AUSDRUCK POSITIONIEREN <135>
100 POKE 214,PO-1:PRINT:POKE 211,PT:PRINT
CHR$(145)PR$:RETURN <110>
198 : <174>
199 REM -----VORBLAETTERN <173>
200 IF II=240 THEN RETURN <137>

```

Listing 1. Super-Install. Bitte mit dem Checksummer 64V3 (Seite 76) eingeben

```

205 II=II+16:FOR I=0 TO 15:POKE 214,I+4:PR
INT:POKE 211,23:PRINT LEFT$(BL$,3) <096>
210 POKE 214,I+4:PRINT:POKE 211,22:PRINT I
N(I+II) <032>
220 POKE(1236+I*40),BI(I+II):POKE(1242+I*4
0),BI(IN(I+II)):NEXT:RETURN <093>
298 : <020>
299 REM -----ZURUECKBLAETTERN <016>
300 IF II<=0 THEN RETURN <056>
305 II=II-16:FOR I=15 TO 0 STEP-1:POKE 214
,I+4:PRINT:POKE 211,23:PRINT LEFT$(BL$,
3) <147>
310 POKE 214,I+4:PRINT:POKE 211,22:PRINT I
N(I+II) <134>
320 POKE(1236+I*40),BI(I+II):POKE(1242+I*4
0),BI(IN(I+II)):NEXT:RETURN <195>
598 : <066>
599 REM -----EDITOR(IDEE:DATA BECKER) <188>
600 IN=0:POKE 214,PO-1:PRINT:POKE 211,PT:P
RINT FR$;" "; <192>
605 POKE 204,0 <252>
609 POKE 212,1:GET IN$:IF IN$=""THEN 609 <216>
610 IF NU=0 THEN 616 <000>
611 IF IN=0 THEN 615 <100>
612 IF IN$=CHR$(20)THEN 620 <080>
613 IF IN$=CHR$(17)OR IN$=CHR$(145)THEN PO
KE 631,ASC(IN$):POKE 198,1:IN$=CHR$(13
):GOTO 625 <137>
614 IF IN$="S"OR IN$="L"OR IN$=" "THEN POK
E 631,ASC(IN$):POKE 198,1:IN$=CHR$(13)
:NU=0:GOTO 625 <151>
615 :IF ASC(IN$)<48 OR ASC(IN$)>57 THEN 60
5 <101>
616 IF IN$=CHR$(29)OR IN$=CHR$(157)THEN IN
$="":GOTO 605 <089>
617 IF IN$=" "OR IN$=CHR$(145)OR IN$=CHR$(
17)THEN POKE 204,1:POKE 212,0:RETURN <026>
618 IF IN$=CHR$(13)AND IN=0 THEN IN$="":GO
TO 605 <203>
620 POKE 207,0 <107>
625 IF IN$=CHR$(13)THEN POKE 204,1:PRINT C
HR$(32):IN$="":GOTO 650 <137>
630 IF IN$=CHR$(20)THEN GOSUB 660:GOTO 605 <117>
635 IF IN>LL-1 THEN 605 <079>
640 PRINT IN$; <106>
645 IN$(IN)=IN$:IN=IN+1:GOTO 605 <074>
650 FOR SL=0 TO IN-1 <194>
655 IN$=IN$+IN$(SL):NEXT:RETURN <236>
660 IF IN=0 THEN RETURN <183>
665 POKE 212,0:POKE 207,0:PRINT CHR$(157)C
HR$(32)CHR$(32)CHR$(157)CHR$(157); <238>
670 IN=IN-1:IN$(IN)="" :RETURN <140>
698 : <166>
699 REM -----ANZEIGE <240>
700 IF ZE>18 THEN ZE=3:POKE 198,0:PO=21:PT
=26:PR$=CHR$(32):GOSUB 100 <168>
705 ZE=ZE+1:PT=26:POKE 214,ZE-1:PRINT:POKE
211,PT:PRINT CHR$(32)CHR$(158); <197>
715 PRINT CHR$(157)CHR$(17)CHR$(18)CHR$(32
)CHR$(146)CHR$(159):IF ZE=4 THEN GOSUB
200 <130>
720 RETURN <014>
740 : <208>
750 IF ZE<5 THEN ZE=20:POKE 198,0:PO=6:PT=
26:PR$=CHR$(32):GOSUB 100 <033>
755 ZE=ZE-1:PT=26:POKE 214,ZE+1:PRINT:POKE
211,PT:PRINT CHR$(32)CHR$(158); <247>
765 PRINT CHR$(157)CHR$(145)CHR$(18)CHR$(3
2)CHR$(146)CHR$(159):IF ZE=19 THEN GOS
UB 300 <225>
770 RETURN <066>
798 : <012>
799 REM -----EDITIEREN <247>
800 FR$="":PO=ZE+1:PT=22:LL=3:NU=1:POKE 21
4,PO-1:PRINT:POKE 211,23:PRINT LEFT$(B
L$,4) <249>
810 POKE 631,ASC(X$):POKE 198,1:GOSUB 600:
IF VAL(IN$)>255 THEN X$=CHR$(32):GOTO
800 <146>
820 IN(II+ZE-4)=VAL(IN$):POKE 1242+(ZE-4)*
40,BI(VAL(IN$)) <239>
830 RETURN <126>
898 : <112>
899 REM -----HAUPTMENUE LAYOUT <243>

```

```

900 POKE 214,0:PRINT:POKE 211,7:FOR Q=1 TO 25:PRINT CHR$(175);:NEXT <182>
905 PR$="ZEICHEN(4SPACE)CODE":PO=3:PT=12:GOSUB 100 <129>
909 POKE 214,1:PRINT:FOR Q=1 TO 21:POKE 211,7:PRINT CHR$(165):POKE 211,31:PRINT CHR$(145)CHR$(167):NEXT <061>
910 PR$="TTTTTTT(4SPACE)TTTT":PO=4:PT=12:GOSUB 100 <009>
915 PR$=CHR$(18)+" SICHERN = LADEN = ENDE":PO=23:PT=7:GOSUB 100:PRINT BL$ <235>
916 RETURN <212>
917 : <131>
918 REM -----HAUPTMENUE ABFRAGEN <188>
919 GOSUB 900:II=-16:GOSUB 200:GOSUB 2350:REM VORGABEN LISTEN <241>
920 GET X$:IF X$="" THEN 920 <228>
925 IF ASC(X$)>47 AND ASC(X$)<58 THEN GOSUB 800:GOTO 920 <110>
930 IF X$<>"S"AND X$<>"L"AND X$<>CHR$(17)AND X$<>CHR$(145)AND X$<>"+"AND X$<>CHR$(24) THEN 920 <178>
940 IF X$=CHR$(17)AND II+ZE<259 THEN GOSUB 700:GOTO 920 <222>
945 IF X$=CHR$(17)AND II+ZE>258 THEN II=-16:GOSUB 700:GOTO 920 <145>
950 IF X$=CHR$(145)AND II+ZE>4 THEN GOSUB 750:GOTO 920 <098>
955 IF X$=CHR$(145)AND II+ZE<5 THEN II=256:GOSUB 750:GOTO 920 <164>
960 IF X$="S" THEN GOSUB 1100:PO=24:PT=0:PR$=BL$:GOSUB 100:GOTO 920 <098>
970 IF X$="L" THEN GOSUB 1000:PO=24:PT=0:PR$=BL$:GOSUB 100:GOTO 920 <217>
980 IF X$=CHR$(24) THEN GOSUB 3000 <116>
985 IF X$="+" THEN POKE 788,49:PRINT CHR$(147):END <021>
990 GOTO 920 <252>
998 : <212>
999 REM -----INDEXFELD LADEN <102>
1000 FR$="CODE LADEN":LL=16:PO=23:PT=7:NU=0:GOSUB 600:N$=IN$ <227>
1002 IF IN$="+"OR IN$=CHR$(145)OR IN$=CHR$(17) THEN POKE 631,ASC(IN$):POKE 198,1:RETURN <244>
1005 OPEN 15,8,15:OPEN 1,8,2,IN$+",P,R":GOSUB 2200:REM FEHLER <163>
1010 IF FE<>0 THEN 1090 <059>
1020 GET#1,Z$,V$:AD=ASC(Z$+CHR$(0))+256*ASC(V$+CHR$(0)):REM ADRESSE <078>
1030 FOR I=0 TO 36:GET#1,X$:NEXT:REM UEBERLESEN <226>
1040 FOR I=0 TO 255:GET#1,X$:IN(I)=ASC(X$+CHR$(0)):NEXT <078>
1045 GET#1,L$:LE=ASC(L$+CHR$(0))-1 <030>
1050 TX$="":FOR I=0 TO LE:GET#1,X$:TX$=TX$+X$:NEXT:GOSUB 2350:REM TEXTPRG.NAME <195>
1060 II=-16:GOSUB 200:REM NEU LISTEN <053>
1090 CLOSE 1:CLOSE 15:IN$="":RETURN <186>
1098 : <058>
1099 REM -----INDEXFELD SICHERN <066>
1100 FR$="CODE SICHERN":LL=16:PO=23:PT=7:NU=0:GOSUB 600:N$=IN$ <172>
1101 GOSUB 1190:IF PEEK(198)=1 THEN RETURN <167>
1102 FR$="ADRESSE <DEZ>":LL=5:GOSUB 600:AD=VAL(IN$):GOSUB 1190:IF PEEK(198)=1 THEN RETURN <204>
1103 GOSUB 2400:AA=AD:GOSUB 2450:IF AD=0 THEN 1102 <188>
1104 FR$="TEXTPROGRAMM":LL=16:GOSUB 600:TX$=IN$:GOSUB 1190:IF PEEK(198)=1 THEN RETURN <089>
1110 OPEN 15,8,15:OPEN 1,8,1,"@: "+N$:GOSUB 2200:REM FEHLER <138>
1112 IF FE<>0 THEN CLOSE 1:CLOSE 15:IN$="":RETURN <106>
1115 GOSUB 2350:REM STATUSZEILE <222>
1120 PRINT#1,CHR$(LB);CHR$(HB);:REM ADRESSE <006>
1125 K=0:FOR I=AD TO (AD+36):PRINT#1,CHR$(DT(K));:K=K+1:NEXT:REM MASCHINENPRG DRUCKER <089>
1130 K=0:FOR I=(AD+36) TO (AD+36+255):PRINT#1,CHR$(IN(K));:K=K+1:NEXT:REM INDEXFELD
LD <158>
1135 PRINT#1,CHR$(LEN(TX$)); <222>
1140 K=1:FOR I=(AD+36+255) TO (AD+36+255+LEN(TX$)-1) <017>
1145 PRINT#1,CHR$(ASC(MID$(TX$,K,1)));:K=K+1:NEXT:REM NAME <228>
1150 CLOSE 1:CLOSE 15:GOSUB 3000:IN$="":RETURN <136>
1188 : <148>
1189 REM -----ABFRAGE AUSSTIEG <134>
1190 PO=24:PT=0:PR$=BL$:GOSUB 100 <010>
1191 IF IN$="+"OR IN$=CHR$(145)OR IN$=CHR$(17) THEN POKE 631,ASC(IN$):POKE 198,1:RETURN <179>
1192 PO=23:PT=7:RETURN <180>
2198 : <142>
2199 REM -----FEHLER <111>
2200 INPUT#15,FE,FE$:IF FE=0 THEN 2220 <057>
2210 PR$=BL$:PO=24:PT=0:GOSUB 100:PR$=FE$:PT=(40-LEN(FE$))/2:GOSUB 100:GOSUB 2300 <112>
2220 RETURN <246>
2298 : <242>
2299 REM -----WARTESCHLEIFE <101>
2300 FOR Q=1 TO 2500:NEXT:RETURN <103>
2348 : <038>
2349 REM -----STATUSZEILE <053>
2350 PO=1:PT=0:PR$=BL$:GOSUB 100:AD$=RIGHT$(STR$(AD),LEN(STR$(AD))-1) <173>
2352 PR$=CHR$(30)+CHR$(34)+N$+CHR$(47)+AD$+CHR$(47)+TX$+CHR$(34)+CHR$(159) <006>
2353 IF LEN(PR$)>40 THEN PR$=LEFT$(PR$,39)+CHR$(34) <118>
2355 PT=INT((41-LEN(PR$))/2):GOSUB 100:RETURN <061>
2398 : <088>
2399 REM -----ADRESSE ANPASSEN <049>
2400 AA=AD+11:GOSUB 2450:DT(1)=LB:DT(6)=HB <173>
2405 AA=AD+35:GOSUB 2450:DT(20)=LB:DT(21)=HB:DT(28)=LB:DT(29)=HB <162>
2410 AA=AD+37:GOSUB 2450:DT(24)=LB:DT(25)=HB <242>
2420 RETURN <192>
2448 : <138>
2449 REM -----HIGH/LOW BYTE <168>
2450 HB=INT(AA/256):LB=AA-HB*256:RETURN <055>
2998 : <180>
2999 REM -----LADEPROGRAMM GENERIEREN <088>
3000 PO=23:PT=7:PR$=CHR$(18)+"MIT CTRL-X LADER ERZEUGEN":GOSUB 100 <148>
3002 PO=24:PR$="PROGRAMM WIRD GELOESCHT !":GOSUB 100 <000>
3003 GET X$:IF X$="" THEN 3003 <009>
3005 IF X$<>CHR$(24) THEN GOSUB 915:RETURN <099>
3010 PRINT CHR$(147);LA$(0) <141>
3020 PRINT:PRINT:PRINT LA$(1);CHR$(34);N$;CHR$(34);LA$(2):PRINT LA$(3) <098>
3030 PRINT LA$(4);CHR$(34);LA$(5);STR$(AD);LA$(6);CHR$(34);LA$(7);CHR$(34);TX$; <225>
3040 PRINT CHR$(34);LA$(7);CHR$(34);LA$(2):PRINT LA$(8) <070>
3050 PRINT LA$(9);CHR$(34);LA$(10)+TX$;CHR$(34);LA$(11) <254>
3060 POKE 631,19:FOR I=632 TO 637:POKE I,13:NEXT:POKE 198,7:END <207>
4998 : <148>
4999 REM -----DRUCKERPROGRAMM <185>
5000 DATA 169,11,141,38,3,169,192,141,39,3,96,72,165,154,201,4,208,12,104,140 <174>
5010 DATA 35,192,168,185,37,192,72,172,35,192,104,76,202,241,255,255,255 <014>
5998 : <132>
5999 REM -----DATAS-LADEPROGRAMM <034>
6000 DATA "NEW", "1 PRINTCHR$(147):IFA=0T HENA=1:LOAD" <023>
6010 DATA ",8,1" <085>
6020 DATA "3 PRINTCHR$(19);", "5 PRINT", "SYS", ":LOAD", "CHR$(34)" <169>
6030 DATA "7 POKE631,19:POKE632,13:POKE198,2:END" <044>
6040 DATA "SAVE", "@: LADER.", ",8" <028>

```

Listing 1. Super-Install (Schluß)